

Министерство образования и науки Российской Федерации
Санкт-Петербургский политехнический университет Петра Великого

Институт компьютерных наук и кибербезопасности
Высшая школа технологий искусственного интеллекта
Направление: 02.03.01 «Математика и компьютерные науки»

Отчёт о выполнении курсовой работы

По дисциплине: «Теория графов»

На тему: «Выбор параметров шрифтовой гарнитуры для
печати бейждей на офсетном принтере»

Выполнил студент
группы 5130201/40002

_____ Семенов И. А.

Проверил
преподаватель

_____ Востров А. В.

«_____» _____ 2026г.

Содержание

Введение	3
1 Постановка задачи	4
2 Предметная область	5
3 Математическое описание	7
3.1 Экспертные системы	7
3.2 Структура экспертной системы	7
3.3 Продукционная модель представления знаний	8
3.4 Бинарное дерево решений	9
3.5 Механизм логического вывода	12
4 Особенности реализации	13
4.1 Среда реализации	13
4.2 Шаблоны	13
4.3 Функции	13
4.4 Факты	15
5 Результаты работы	17
Заключение	21
Список литературы	22
Приложение. Реализация экспертной системы в среде CLIPS	23

Введение

В отчёте представлено описание выполнения курсовой работы по дисциплине «Теория графов». В работе требуется разработать экспертную систему, основанную на бинарном дереве решений и продукционной модели представления знаний.

Разрабатываемая система предназначена для выбора параметров шрифтовой гарнитуры при подготовке бейджей к офсетной печати. Пользователь отвечает на вопросы о составе макета, алфавите, условиях чтения и технологических особенностях печати, после чего система выдаёт набор рекомендаций по выбору гарнитуры, насыщенности, контраста и межбуквенного интервала.

В отчёте рассмотрены предметная область, математическая модель экспертной системы, структура дерева решений, особенности реализации и демонстрация работы программы на нескольких сценариях.

1 Постановка задачи

Цель работы: разработать экспертную систему для выбора параметров шрифтовой гарнитуры при подготовке бейджей к офсетной печати. Система должна задавать пользователю вопросы о макете и условиях печати, а затем выдавать типографические рекомендации.

Задачи.

1. Выделить критерии, влияющие на выбор параметров шрифта.
2. Построить бинарное дерево решений. Минимальный размер дерева: 4 яруса, 30 узлов.
3. Реализовать модель в среде CLIPS.

2 Предметная область

В качестве предметной области выбран подбор параметров шрифтовой гарнитуры для бейджей, изготавливаемых методом офсетной печати. Бейдж имеет ограниченную площадь, но при этом должен быстро считываться пользователем: на нём обычно размещаются имя, логотип организации и, при необходимости, должность, подразделение или дополнительная служебная информация.

Основная задача подбора шрифта заключается в том, чтобы сохранить читаемость текста при заданных условиях просмотра и печати. На читаемость влияют размер символов, контраст текста и фона, насыщенность начертания, форма знаков, межбуквенный интервал и различимость тонких элементов при визуальном восприятии.

Основой для выбора критериев служит стандарт ISO 24509:2019, описывающий оценку минимального читаемого размера символов с учётом условий просмотра, в том числе дистанции чтения, яркости и контраста [2]. Для задачи печати бейджей эти факторы были адаптированы к практическим признакам макета и материала. В системе учитываются следующие критерии:

1. наличие дополнительного текста помимо имени и логотипа;
2. наличие кириллицы в текстовом содержимом;
3. необходимость чтения бейджа с расстояния более одного метра;
4. использование тёмного фона;
5. наличие ламинации;
6. плотность бумаги выше 250 г/м^2 .

Наличие дополнительного текста повышает требования к разборчивости гарнитуры, поскольку на той же площади нужно разместить больше информации. Кириллица учитывается отдельно, так как кириллические символы сложнее считывать с увеличенной дистанции. Увеличенная дистанция чтения требует более выраженных форм символов, большей высоты строчных знаков и повышенного контраста. Тёмный фон и ламинация дополнительно усложняют восприятие текста: блики и снижение контраста делают тонкие штрихи менее надёжными. Высокая плотность бумаги, наоборот, позволяет использовать более аккуратные детали.

Множество возможных решений системы образуют рекомендации по следующим параметрам шрифта:

- тип гарнитуры:

- гротеск;
- гуманистический гротеск;
- антиква;
- насыщенность начертания:
 - Regular;
 - SemiBold;
 - Bold;
- допустимость тонких элементов:
 - допускаются;
 - допускаются умеренно;
 - следует избегать;
- высота строчных знаков:
 - стандартная;
 - увеличенная;
- межбуквенный интервал:
 - стандартный;
 - слегка увеличенный;
 - увеличенный;
- контраст текста и фона:
 - достаточный;
 - высокий.

3 Математическое описание

3.1 Экспертные системы

Экспертная система — это программная система, моделирующая рассуждения специалиста в некоторой узкой предметной области. В отличие от традиционной программы, которая обычно реализует заранее известный алгоритм, экспертная система применяется для задач, где полный алгоритм решения трудно или невозможно задать заранее.

Экспертная система обычно включает интерфейс пользователя, базу знаний и механизм логического вывода. Она работает в двух основных режимах: режиме приобретения знаний, когда база знаний наполняется сведениями предметной области, и режиме решения задачи, когда пользователь отвечает на вопросы системы и получает итоговый результат.

3.2 Структура экспертной системы

Структура экспертной системы включает интерфейс пользователя, подсистему вывода, базу знаний, данные, модель представления данных и интеллектуальный редактор (рис. 1).




Рис. 1: Структура экспертной системы

Знания — это правила, законы, закономерности, полученные в результате профессиональной деятельности в пределах предметной области.

База знаний — база данных, содержащая правила вывода и информацию о человеческом опыте и знаниях в некоторой предметной области.

Другими словами, это набор таких закономерностей, которые устанавливают связи между вводимой и выводимой информацией.

Данные — это совокупность фактов и идей, представленных в формализованном виде.

Модель представления данных — это способ задания знаний для их хранения, удобного доступа и взаимодействия с ними, который подходит под задачу интеллектуальной системы.

3.3 Продукционная модель представления знаний

Продукция — это правило, задающее связь между исходной ситуацией и выводимым результатом. Условие правила называется антецедентом, а действие — консеквентом.

Продукционная система включает в себя три компоненты:

1. набор правил (продукций), представляющих базу знаний;
2. рабочую память, в которой хранятся исходные факты и факты, полученные в процессе вывода;
3. механизм логического вывода, применяющий правила к имеющимся фактам.

В основе продукционной модели находится правило вида:

ЕСЛИ $\langle \text{условие} \rangle$, ТО $\langle \text{действие}_1 \rangle$, ИНАЧЕ $\langle \text{действие}_2 \rangle$.

Если условие выполняется, применяется первое действие. Если условие не выполняется, выполняется альтернативное действие. Условия могут объединяться с помощью логических связок AND и OR. Правило сработает, если при совпадении фактов из рабочей памяти обнаруживается совпадение с акцентом правила. После этого результат заносится в рабочую память.

Структура продукционного правила показана на рис. 2.




Рис. 2: Структура продукционного правила

Достоинством продукционной модели является простота добавления и чтения отдельных правил. Каждое правило можно рассматривать независимо, поэтому небольшая экспертная система остаётся понятной и легко расширяется. Недостатки проявляются при росте числа правил: становится сложнее контролировать полноту базы знаний, пересечения условий и общий порядок вывода.


3.4 Бинарное дерево решений

Бинарное дерево решений представляет собой ориентированное дерево, в котором каждый внутренний узел имеет два исходящих ребра по ветвям «Да» и «Нет». Использование древовидной структуры позволяет ограничить число проверяемых правил длиной пути от корня до листа.

Полное дерево решений приведено на рисунке 3. На таблице 1 представлены его параметры.

Таблица 1: Параметры бинарного дерева решений

Название	Значение
Количество правил	15
Количество фактов	16
Общее число узлов	31
Минимальное число правил до результата	3
Максимальное число правил до результата	5
Количество уровней дерева	6



3.5 Механизм логического вывода

Механизм логического вывода данных выполняет анализ имеющихся фактов и применяет к ним правила из базы знаний для получения нового факта или конечного результата.

В разработанной экспертной системе логический вывод выполняется прямым способом: правила сопоставляются с фактами рабочей памяти, после чего активное правило либо задаёт следующий вопрос, либо выводит итоговую рекомендацию. Алгоритм работы системы следующий:

1. В рабочей памяти отсутствует факт о первом критерии, поэтому активируется корневое правило.
2. Система выводит пользователю текст вопроса и два варианта ответа: «Нет» и «Да».
3. Пользователь вводит номер выбранного варианта.
4. Система проверяет корректность ввода. Если введено не 1 и не 2, запрос повторяется.
5. По корректному ответу в рабочую память добавляется факт со значением **yes** или **no**.
6. Механизм вывода снова сопоставляет факты рабочей памяти с правилами базы знаний и выбирает правило, условия которого стали истинными.
7. Если выбранное правило является правилом, процесс повторяется со второго шага.
8. Если выбранное правило соответствует конечной вершине дерева решений, система выводит итоговую рекомендацию по параметрам шрифтовой гарнитуры и завершает работу.

4 Особенности реализации

4.1 Среда реализации

Экспертная система реализована в среде CLIPS. Эта среда подходит для выбранной продукционной модели, так как позволяет описывать знания через правила, хранить ответы пользователя в рабочей памяти и запускать прямой логический вывод. Математическая модель дерева решений описана в разделе 3.4, а в данном разделе рассматривается её программная реализация.

4.2 Шаблоны

Рабочая память системы хранит ответы пользователя в однотипных фактах `known`. Структура такого факта приведена в листинге 1: поле `attribute` задаёт имя проверяемого признака, а поле `value` хранит бинарное значение `yes` или `no`.

Листинг 1: Шаблон факта `known`

```
1 (deftemplate known
2   (slot attribute (type SYMBOL))
3   (slot value (type SYMBOL))
4 )
```

4.3 Функции

Для обработки ввода используется функция `ask-choice-1-2`, приведённая в листинге 2. Функция повторяет запрос до тех пор, пока пользователь не введёт допустимое значение. Такой вариант выбран вместо свободного ввода строк, чтобы избежать неоднозначных ответов и опечаток.

Вход: значение, вводимое пользователем с клавиатуры.

Выход: число 1, если пользователь выбрал отрицательный ответ, или число 2, если пользователь выбрал положительный ответ.

Листинг 2: Функция проверки бинарного выбора

```
7 (deffunction ask-choice-1-2 ()
8   (bind ?ans 0)
9
10  (while (and (neq ?ans 1) (neq ?ans 2)) do
11    (printout t "Ваш выбор (введите 1 или 2): ")
12    (bind ?ans (read))
13
14    (if (and (neq ?ans 1) (neq ?ans 2))
15      then
16        (printout t crlf "Ошибка! Можно ввести только 1 или 2.
17          ↪ Попробуйте ещё раз." crlf)
18    )
19
20  (return ?ans)
```

Универсальная функция вопроса `ask-binary` показана в листинге 3. Для получения ответа пользователя функция вызывает проверку выбора из листинга 2.

Вход: имя проверяемого признака, текст вопроса и выбранный пользователем вариант ответа.

Выход: факт с выбранным значением, добавленный в рабочую память.

Листинг 3: Функция задания бинарного вопроса

```
24 (deffunction ask-binary (?attribute ?question)
25   (printout t crlf)
26   (printout t ?question crlf)
27   (printout t "1) Нет" crlf)
28   (printout t "2) Да" crlf)
29
30   (bind ?ans (ask-choice-1-2))
31
32   (if (= ?ans 1)
33     then
34       (assert (known (attribute ?attribute) (value no)))
35     else
36       (assert (known (attribute ?attribute) (value yes)))
37   )
38 )
```

Функция `print-result`, представленная в листинге 4, используется всеми листовыми правилами.

Вход: номер листа и список рекомендаций.

Выход: текст результата, напечатанный в консоль; после вывода работа системы завершается.

Листинг 4: Функция вывода результата

```
41 (deffunction print-result (?number $?items)
42   (printout t crlf "Лист " ?number crlf)
43
44   (foreach ?item ?items
45     (printout t "- " ?item crlf)
46   )
47
48   (halt)
49 )
50
```

4.4 Факты

Факты создаются только после ответа пользователя на вопрос. Например, если пользователь подтверждает наличие кириллицы, функция из листинга 3 добавляет факт (`known (attribute cyrillic) (value yes)`). Если ответ отрицательный, создаётся аналогичный факт со значением `no`.

Такая организация делает правила независимыми от порядка ввода: каждое правило проверяет только те факты, которые нужны для соответствующей ветви дерева. Пример правила-вопроса приведён в листинге 5. Оно срабатывает только после подтверждения наличия дополнительного текста и задаёт следующий вопрос о кириллице.

Листинг 5: Правило-вопрос о наличии кириллицы

```
62 (defrule ask-cyrillic
63   (known (attribute additional-text) (value yes))
64   (not (known (attribute cyrillic)))
65   =>
66   (ask-binary cyrillic
67     "Есть ли кириллица?")
68 )
```

Листовые правила проверяют полное сочетание признаков и вызывают функцию вывода результата из листинга 4. Пример такого правила приведён в листинге 6; оно соответствует одному из конечных листьев дерева на рисунке 3.

Листинг 6: Пример листового правила

```
361 (defrule node-16
362   (known (attribute additional-text) (value yes))
363   (known (attribute cyrillic) (value yes))
364   (known (attribute read-over-1m) (value yes))
365   (known (attribute dark-background) (value yes))
366   (known (attribute lamination) (value yes))
367   =>
368   (print-result 16
369     "гротеск"
370     "кириллица"
371     "Bold"
372     "увеличенная высота строчных знаков"
373     "высокий контраст текста и фона"
374     "избегать тонких штрихов"
375     "увеличенный межбуквенный интервал")
376 )
```


5 Результаты работы

Для проверки работы экспертной системы были выполнены несколько консультационных сценариев. В каждом сценарии пользователь отвечает на вопросы системы, после чего программа выводит номер листа дерева решений и набор типографических рекомендаций.

На рисунке 4 показан сценарий, в котором дополнительный текст не требуется, фон бейджа светлый, ламинация отсутствует. Система переходит в лист 1 и рекомендует использовать гротеск или антикву обычной либо полужирной насыщенности, сохраняя стандартный межбуквенный интервал и достаточный контраст текста и фона.




Рис. 4: Демонстрация работы системы для листа 1

На рисунке 5 приведён сценарий с дополнительным кириллическим текстом, чтением с расстояния более одного метра, тёмным фоном и ламинацией. В этом случае система выбирает лист 16 и усиливает требования к читаемости: рекомендует гротеск, жирное начертание, увеличенную высоту строчных знаков, высокий контраст и увеличенный межбуквенный интервал.




Рис. 5: Демонстрация работы системы для листа 16

На рисунке 6 показан сценарий с дополнительным латинским текстом, чтением с расстояния более одного метра и тёмным фоном без ламинации. Система переходит в лист 9 и рекомендует гротеск, полужирное начертание, увеличенную высоту строчных знаков и высокий контраст текста и фона.




Рис. 6: Демонстрация работы системы для листа 9

На рисунке 7 показана обработка некорректного пользовательского ввода. Если пользователь вводит значение, отличное от 1 или 2, система выводит сообщение об ошибке и повторяет вопрос, не переходя к следующему узлу дерева.

```
Dir: ~/poly/4sem/graphs/coursework
Ваш выбор (введите 1 или 2): 1

Лист 9
- гротеск
- латиница
- SemiBold
- увеличенная высота строчных знаков
- высокий контраст текста и фона
CLIPS> (reset)
CLIPS> (run)

Требуется ли размещение дополнительного текста помимо имени и логотипа?
1) Нет
2) Да
Ваш выбор (введите 1 или 2): 1

Фон бейджа тёмный?
1) Нет
2) Да
Ваш выбор (введите 1 или 2): 3

Ошибка! Можно ввести только 1 или 2. Попробуйте ещё раз.
Ваш выбор (введите 1 или 2): 2

Есть ли ламинация?
1) Нет
2) Да
Ваш выбор (введите 1 или 2): 5

Ошибка! Можно ввести только 1 или 2. Попробуйте ещё раз.
Ваш выбор (введите 1 или 2): 1

Лист 3
- гротеск
- SemiBold / Bold
- увеличенная высота строчных знаков
- высокий контраст текста и фона
CLIPS>
```

Рис. 7: Обработка некорректного ввода

Во всех приведённых сценариях система корректно проходит по ветвям дерева решений, задаёт вопросы только по выбранному пути, обрабатывает ошибочный ввод и завершает работу выводом рекомендаций, соответствующих достигнутому листу.

Заключение

В ходе выполнения курсовой работы была разработана экспертная система для выбора параметров шрифтовой гарнитуры при подготовке бейджей к офсетной печати. Была описана предметная область, выделены основные критерии выбора, построено бинарное дерево решений и выполнена реализация модели в среде CLIPS. Дерево содержит 15 правил, 16 фактов и 6 ярусов. Работа системы продемонстрирована на нескольких сценариях, в которых программа задаёт пользователю вопросы и выдаёт рекомендации.

Достоинства реализации

- База знаний опирается на рекомендации ISO 24509:2019 по читаемости текста и охватывает наиболее частые сценарии подготовки бейджей.
- Сложность алгоритма составляет $O(\log n)$, где n — число конечных решений.

Недостатки реализации

- При добавлении новых правил может потребоваться перестроение поддерева, корнем которого является изменяемый узел-правило.
- Система не допускает неопределённых ответов пользователя (например, «не знаю»), поэтому для получения результата необходимо дать однозначный ответ на каждый задаваемый вопрос.

Масштабируемость

- Добавление графического интерфейса.
- Упрощение процесса добавление новых знаний, например, через графический интерфейс.

Таким образом, разработанная экспертная система выполняет поставленную задачу и может быть расширена для более детальной типографической консультации.

Список литературы

- [1] Новиков Ф. А. Дискретная математика для программистов: Учебник для вузов. 3-е изд. — СПб.: Питер, 2009. — 384 с.
- [2] ISO 24509:2019. Ergonomics — Accessible design — A method for estimating minimum legible font size for people at any age [Электронный ресурс]. URL: <https://cdn.standards.iteh.ai/samples/69766/d14fe0798a1b48f9ba4a13509d1d5d12/ISO-24509-2019.pdf> (дата обращения: 17.05.2026).
- [3] Экспертные системы: презентация по дисциплине «Теория графов» [Электронный ресурс]. URL: https://tema.spbstu.ru/userfiles/files/courses/2019-theory_graph/ES_TG.pdf (дата обращения: 11.05.2026).
- [4] CLIPS Reference Manual [Электронный ресурс]. URL: <https://www.clipsrules.net/Documentation.html> (дата обращения: 28.05.2026).

Приложение. Реализация экспертной системы в среде CLIPS

```
1 (deftemplate known
2   (slot attribute (type SYMBOL))
3   (slot value (type SYMBOL))
4 )
5
6 ; обработка некорректного ввода
7 (deffunction ask-choice-1-2 ()
8   (bind ?ans 0)
9
10   (while (and (neq ?ans 1) (neq ?ans 2)) do
11     (printout t "Ваш выбор (введите 1 или 2): ")
12     (bind ?ans (read))
13
14     (if (and (neq ?ans 1) (neq ?ans 2))
15       then
16         (printout t crlf "Ошибка! Можно ввести только 1 или 2. Попробуйте ещё раз." crlf)
17     )
18   )
19
20   (return ?ans)
21 )
22
23 ; универсальная функция вопроса
24 (deffunction ask-binary (?attribute ?question)
25   (printout t crlf)
26   (printout t ?question crlf)
27   (printout t "1) Нет" crlf)
28   (printout t "2) Да" crlf)
29
30   (bind ?ans (ask-choice-1-2))
31
32   (if (= ?ans 1)
33     then
34       (assert (known (attribute ?attribute) (value no)))
35     else
36       (assert (known (attribute ?attribute) (value yes)))
37   )
38 )
39
40 ; функция вывода результата
41 (deffunction print-result (?number $?items)
42   (printout t crlf "Лист " ?number crlf)
43
44   (foreach ?item ?items
45     (printout t "- " ?item crlf)
46   )
47
48   (halt)
49 )
50
51 ; -----
52 ; ВОПРОСЫ
53 ; -----
54
55 (defrule ask-additional-text
56   (not (known (attribute additional-text)))
57   =>
58   (ask-binary additional-text
59     "Требуется ли размещение дополнительного текста помимо имени и логотипа?")
60 )
61
62 (defrule ask-cyrillic
63   (known (attribute additional-text) (value yes))
64   (not (known (attribute cyrillic)))
65   =>
66   (ask-binary cyrillic
67     "Есть ли кириллица?")
68 )
69
70 (defrule ask-reading-distance
71   (known (attribute additional-text) (value yes))
72   (known (attribute cyrillic))
```

```

73     (not (known (attribute read-over-1m)))
74     =>
75     (ask-binary read-over-1m
76       "Требуется ли чтение с расстояния более 1 метра?")
77   )
78
79   (defrule ask-dark-background-without-additional-text
80     (known (attribute additional-text) (value no))
81     (not (known (attribute dark-background)))
82     =>
83     (ask-binary dark-background
84       "Фон бейджа тёмный?")
85   )
86
87   (defrule ask-dark-background-with-additional-text
88     (known (attribute additional-text) (value yes))
89     (known (attribute read-over-1m) (value yes))
90     (not (known (attribute dark-background)))
91     =>
92     (ask-binary dark-background
93       "Фон бейджа тёмный?")
94   )
95
96   (defrule ask-lamination-without-additional-text
97     (known (attribute additional-text) (value no))
98     (known (attribute dark-background))
99     (not (known (attribute lamination)))
100    =>
101    (ask-binary lamination
102      "Есть ли ламинация?")
103  )
104
105  (defrule ask-lamination-with-additional-text
106    (known (attribute additional-text) (value yes))
107    (known (attribute read-over-1m) (value yes))
108    (known (attribute dark-background))
109    (not (known (attribute lamination)))
110    =>
111    (ask-binary lamination
112      "Есть ли ламинация?")
113  )
114
115  (defrule ask-paper-density
116    (known (attribute additional-text) (value yes))
117    (known (attribute read-over-1m) (value no))
118    (not (known (attribute paper-density-over-250)))
119    =>
120    (ask-binary paper-density-over-250
121      "Плотность бумаги выше 250 г/м²?")
122  )
123
124  ; -----
125  ; ЛИСТЫ 1-4
126  ; Нет дополнительного текста
127  ; -----
128
129  (defrule node-1
130    (known (attribute additional-text) (value no))
131    (known (attribute dark-background) (value no))
132    (known (attribute lamination) (value no))
133    =>
134    (print-result 1
135      "гротеск или антиква"
136      "Regular / SemiBold"
137      "допускаются более тонкие элементы"
138      "стандартный межбуквенный интервал"
139      "достаточный контраст текста и фона")
140  )
141
142  (defrule node-2
143    (known (attribute additional-text) (value no))
144    (known (attribute dark-background) (value no))
145    (known (attribute lamination) (value yes))
146    =>
147    (print-result 2
148      "гротеск или антиква"
149      "SemiBold"
150      "увеличенная высота строчных знаков")

```



```

151         "достаточный контраст текста и фона")
152     )
153
154 (defrule node-3
155     (known (attribute additional-text) (value no))
156     (known (attribute dark-background) (value yes))
157     (known (attribute lamination) (value no))
158     =>
159     (print-result 3
160         "гротеск"
161         "SemiBold / Bold"
162         "увеличенная высота строчных знаков"
163         "высокий контраст текста и фона")
164     )
165
166 (defrule node-4
167     (known (attribute additional-text) (value no))
168     (known (attribute dark-background) (value yes))
169     (known (attribute lamination) (value yes))
170     =>
171     (print-result 4
172         "гротеск"
173         "Bold"
174         "увеличенная высота строчных знаков"
175         "высокий контраст текста и фона"
176         "избегать тонких штрихов")
177     )
178
179 ; -----
180 ; ЛИСТЫЯ 5-10
181 ; Дополнительный текст есть, кириллицы нет
182 ; -----
183
184 (defrule node-5
185     (known (attribute additional-text) (value yes))
186     (known (attribute cyrillic) (value no))
187     (known (attribute read-over-1m) (value no))
188     (known (attribute paper-density-over-250) (value no))
189     =>
190     (print-result 5
191         "гротеск"
192         "латиница"
193         "Regular / SemiBold"
194         "простые открытые формы символов"
195         "избегать тонких деталей")
196     )
197
198 (defrule node-6
199     (known (attribute additional-text) (value yes))
200     (known (attribute cyrillic) (value no))
201     (known (attribute read-over-1m) (value no))
202     (known (attribute paper-density-over-250) (value yes))
203     =>
204     (print-result 6
205         "антиква или гротеск"
206         "латиница"
207         "Regular"
208         "допускаются тонкие элементы"
209         "стандартный межбуквенный интервал")
210     )
211
212 (defrule node-7
213     (known (attribute additional-text) (value yes))
214     (known (attribute cyrillic) (value no))
215     (known (attribute read-over-1m) (value yes))
216     (known (attribute dark-background) (value no))
217     (known (attribute lamination) (value no))
218     =>
219     (print-result 7
220         "гротеск"
221         "латиница"
222         "Regular / SemiBold"
223         "увеличенная высота строчных знаков"
224         "высокий контраст текста и фона")
225     )
226
227 (defrule node-8
228     (known (attribute additional-text) (value yes))

```

```

229 (known (attribute cyrillic) (value no))
230 (known (attribute read-over-1m) (value yes))
231 (known (attribute dark-background) (value no))
232 (known (attribute lamination) (value yes))
233 =>
234 (print-result 8
235   "гротеск"
236   "латиница"
237   "SemiBold"
238   "увеличенная высота строчных знаков"
239   "избегать тонких штрихов"
240   "высокий контраст текста и фона")
241 )
242
243 (defrule node-9
244   (known (attribute additional-text) (value yes))
245   (known (attribute cyrillic) (value no))
246   (known (attribute read-over-1m) (value yes))
247   (known (attribute dark-background) (value yes))
248   (known (attribute lamination) (value no))
249   =>
250   (print-result 9
251     "гротеск"
252     "латиница"
253     "SemiBold"
254     "увеличенная высота строчных знаков"
255     "высокий контраст текста и фона")
256 )
257
258 (defrule node-10
259   (known (attribute additional-text) (value yes))
260   (known (attribute cyrillic) (value no))
261   (known (attribute read-over-1m) (value yes))
262   (known (attribute dark-background) (value yes))
263   (known (attribute lamination) (value yes))
264   =>
265   (print-result 10
266     "гротеск"
267     "латиница"
268     "Bold"
269     "увеличенная высота строчных знаков"
270     "высокий контраст текста и фона"
271     "избегать тонких штрихов"
272     "увеличенный межбуквенный интервал")
273 )
274
275 ; -----
276 ; ЛИСТЫЯ 11-16
277 ; Дополнительный текст есть, кириллица есть
278 ; -----
279
280 (defrule node-11
281   (known (attribute additional-text) (value yes))
282   (known (attribute cyrillic) (value yes))
283   (known (attribute read-over-1m) (value no))
284   (known (attribute paper-density-over-250) (value no))
285   =>
286   (print-result 11
287     "гротеск"
288     "кириллица"
289     "SemiBold"
290     "простые открытые формы символов"
291     "увеличенная высота строчных знаков"
292     "избегать тонких деталей"
293     "слегка увеличенный межбуквенный интервал")
294 )
295
296 (defrule node-12
297   (known (attribute additional-text) (value yes))
298   (known (attribute cyrillic) (value yes))
299   (known (attribute read-over-1m) (value no))
300   (known (attribute paper-density-over-250) (value yes))
301   =>
302   (print-result 12
303     "гуманистический гротеск"
304     "кириллица"
305     "SemiBold"
306     "средний или высокий контраст штрихов")

```

```

307     "увеличенная высота строчных знаков"
308     "умеренно увеличенный межбуквенный интервал"
309     "без тонких декоративных элементов")
310 )
311
312 (defrule node-13
313   (known (attribute additional-text) (value yes))
314   (known (attribute cyrillic) (value yes))
315   (known (attribute read-over-1m) (value yes))
316   (known (attribute dark-background) (value no))
317   (known (attribute lamination) (value no))
318   =>
319   (print-result 13
320     "гротеск"
321     "кириллица"
322     "SemiBold"
323     "увеличенная высота строчных знаков"
324     "высокий контраст текста и фона"
325     "слегка увеличенный межбуквенный интервал")
326 )
327
328 (defrule node-14
329   (known (attribute additional-text) (value yes))
330   (known (attribute cyrillic) (value yes))
331   (known (attribute read-over-1m) (value yes))
332   (known (attribute dark-background) (value no))
333   (known (attribute lamination) (value yes))
334   =>
335   (print-result 14
336     "гротеск"
337     "кириллица"
338     "SemiBold / Bold"
339     "увеличенная высота строчных знаков"
340     "высокий контраст текста и фона"
341     "избегать тонких штрихов"
342     "слегка увеличенный межбуквенный интервал")
343 )
344
345 (defrule node-15
346   (known (attribute additional-text) (value yes))
347   (known (attribute cyrillic) (value yes))
348   (known (attribute read-over-1m) (value yes))
349   (known (attribute dark-background) (value yes))
350   (known (attribute lamination) (value no))
351   =>
352   (print-result 15
353     "гротеск"
354     "кириллица"
355     "Bold"
356     "увеличенная высота строчных знаков"
357     "высокий контраст текста и фона"
358     "слегка увеличенный межбуквенный интервал")
359 )
360
361 (defrule node-16
362   (known (attribute additional-text) (value yes))
363   (known (attribute cyrillic) (value yes))
364   (known (attribute read-over-1m) (value yes))
365   (known (attribute dark-background) (value yes))
366   (known (attribute lamination) (value yes))
367   =>
368   (print-result 16
369     "гротеск"
370     "кириллица"
371     "Bold"
372     "увеличенная высота строчных знаков"
373     "высокий контраст текста и фона"
374     "избегать тонких штрихов"
375     "увеличенный межбуквенный интервал")
376 )

```